

Introduction to ODS

Lauren Haworth, Genentech, Inc., South San Francisco, CA

➤ ABSTRACT

Haven't gotten around to reading the entire ODS section in Online Doc? Don't have time to read a whole book on the topic? This presentation will show you the basics of ODS, and then cover some simple tips and tricks from my book "Output Delivery System: The Basics". In no time you'll be impressing your boss and coworkers with beautiful customized ODS output. The presentation will include easy-to-use techniques for enhancing HTML, RTF, and printer results.

This paper is designed for the beginning SAS programmer, and is based on SAS versions 8.2 and higher.

➤ ODS TERMINOLOGY

ODS is a new tool, and comes with some new terminology. When you create ODS output, a number of components are involved. First, there is the procedure output. This is the text and numbers that make up your results. For a simple PROC CORR run on two variables, the procedure output is as follows:

```
ItemPrice  ItemsSold  500  0.04976  0.2667
```

The next component is the table definition. Each procedure has a table definition that tells it how to arrange, display, and label the results. The table definition for PROC CORR tells ODS what to do with the procedure output:

```
1 With Variables:   ItemPrice
1 Variables:       ItemsSold
Pearson Correlation Coefficients, N=500
Prob > |r| under H0: Rho=0
          ItemsSold
ItemPrice      0.04976
Item Price     0.2667
```

The third component of ODS output is the style definition. This defines the appearance of the output, and includes color schemes, fonts, borders, and sizes. The fourth and final component of ODS output is the output destination. Output can be sent to an HTML file, a PDF file, a RTF document, or many other destinations.

So ODS output is created from the procedure output, a table definition, a style definition, and a destination. The power of ODS is that the user has control over all of these aspects. Changing procedure output is easy, and doesn't require any knowledge of ODS. And it is easy to switch style definitions and output destinations. However, changing table definitions requires a lot of expertise, and modifying style definitions is moderately difficult. This paper will focus on selecting styles and output destinations.

➤ SENDING RESULTS TO THE WEB

Creating web output from SAS is easy. As long as you are running version 8.0 or later¹, all you have to add is two lines of code. The first goes right before your reporting procedure:

```
ODS HTML FILE='myfilename.html';
```

¹ ODS was launched with version 7, but that version was not widely released.

After this line of code, you insert the code for your reporting procedure. For example, you could create a table with the TABULATE procedure. Then, after the RUN statement that ends your procedure, you add the following line of code:

```
ODS HTML CLOSE;
```

So the complete code would look like this, with the TABULATE code² in the middle of the “ODS sandwich”.

```
ODS HTML FILE='myfilename.html';
proc tabulate data=census f=dollar8.;
  class sex educ;
  var income;
  table educ='Education',
    income='Average Salary'*
    mean=' *'
    (sex=' ' all);
run;
ODS HTML CLOSE;
```

The ODS HTML result is the output shown below.

| | Average Salary | | |
|-----------------------------|----------------|----------|----------|
| | Male | Female | All |
| Education | | | |
| Not HS graduate | \$15,113 | \$4,449 | \$13,039 |
| HS graduate | \$33,419 | \$17,539 | \$28,464 |
| Some college | \$30,466 | \$22,730 | \$27,514 |
| Associates degree | \$40,690 | \$33,988 | \$38,057 |
| Bachelors degree | \$46,625 | \$43,862 | \$45,821 |
| Post-graduate degree | \$77,195 | \$45,000 | \$60,501 |

If you don't like this look, you can change it by switching styles. The table above uses the default style, which is called “Default”. For a different look try:

```
ODS HTML FILE='myfilename.html'
  STYLE=BarrettsBlue;
  * the TABULATE code goes here ;
ODS HTML CLOSE;
```

| | Average Salary | | |
|-----------------------------|----------------|----------|----------|
| | Male | Female | All |
| Education | | | |
| Not HS graduate | \$15,113 | \$4,449 | \$13,039 |
| HS graduate | \$33,419 | \$17,539 | \$28,464 |
| Some college | \$30,466 | \$22,730 | \$27,514 |
| Associates degree | \$40,690 | \$33,988 | \$38,057 |
| Bachelors degree | \$46,625 | \$43,862 | \$45,821 |
| Post-graduate degree | \$77,195 | \$45,000 | \$60,501 |

² For more information about the TABULATE code, see my paper “SAS Reporting 101,” available at http://www.laurenhawksworth.com/pubs_current.htm.

The output above shows the new look created by the style change from Default to BarrettsBlue. There are over a dozen styles for you to try. And, as you get more experience with ODS, you can even create your own custom style.

For more information on creating HTML output, see my paper “HTML for the SAS Programmer,” available at http://www.laurenhawksworth.com/pubs_current.htm.

➤ COMBINING OUTPUT

The previous example showed how to create a web page from a single SAS procedure. However, ODS is not limited to producing output one procedure at a time. If your report has multiple outputs, with more than one PROC call, you can route all of the output to a single web page. All you need to do is add the additional procedures after the first one, and before the ODS HTML CLOSE statement. For example, you could create a sales report that shows revenue by customer, revenue by division, and the displays detailed data on top performers.

```
title 'Quarterly Report';
ODS HTML FILE='QReport.html';
title2 'Revenue by Customer';
proc means data=Billings nonobs mean sum;
  class CustomerName;
  var BillableAmt;
run;
title2 'Revenue by Division';
proc means data=Billings nonobs mean sum;
  class Division;
  var BillableAmt;
run;
title2 'Top Performers';
proc print data=Totals noobs;
  where BillableAmt>10000;
  var EmployeeName BillableAmt;
run;
ODS HTML CLOSE;
```

This would create the output shown on the right. The PROC PRINT output is not shown, but it would appear below the other two tables. The user would be able to scroll down to see the full report.

Quarterly Report Revenue by Customer

The MEANS Procedure

| Analysis Variable : BillableAmt Amount Billable | | |
|---|-------------|----------|
| Customer | Mean | Sum |
| IPO.com | 365.5745833 | 8773.79 |
| Bricks and Mortar, Inc. | 362.1439130 | 8329.31 |
| Virtual Co. | 309.8741935 | 9606.10 |
| Sweat Shops Athletic | 334.4563333 | 10033.69 |
| Smith & Smith | 296.8818182 | 9764.10 |

Quarterly Report Revenue by Division

The MEANS Procedure

| Analysis Variable : BillableAmt Amount Billable | | |
|---|-------------|----------|
| Division | Mean | Sum |
| Applications | 310.3919048 | 6518.23 |
| Analysis | 289.3442857 | 6055.23 |
| Reporting | 334.8728571 | 7032.33 |
| Systems | 325.7992308 | 6470.78 |
| Documentation | 335.9058333 | 8061.74 |
| QA | 370.3100000 | 10388.68 |

➤ CREATING A TABLE OF CONTENTS

Once you start combining output, you may find that the web page gets large enough that it's hard to find parts of the results. ODS gives you a tool to help users navigate your web reports. You can add a table of contents to your web page. To do this, you need to add two parameters to your ODS HTML call. These set up a frame page and a contents page. The frame page is used to hold your table of contents and your main output. Users open the frame page to see your results. The syntax is simple:

```
ODS HTML FILE = 'QReport.html'
CONTENTS='contents.html'
FRAME='frame.html';
* the procedure calls go here;
ODS HTML CLOSE;
```

The output is shown below:

The screenshot shows a web-based report interface. On the left, there is a sidebar titled "Table of Contents" with three items listed:

1. The Means Procedure
-Summary statistics
2. The Means Procedure
-Summary statistics
3. The Print Procedure
-Data Set
WORK.TOTALS

The main content area has a title "Quarterly Report Revenue by Customer" and a subtitle "The MEANS Procedure". Below this is a table titled "Analysis Variable : BillableAmt Amount Billable". The table has columns for Customer, Mean, and Sum. The data is as follows:

| Customer | Mean | Sum |
|-------------------------|-------------|----------|
| IPO.com | 365.5745833 | 8773.79 |
| Bricks and Mortar, Inc. | 362.1439130 | 8329.31 |
| Virtual Co. | 309.8741935 | 9606.10 |
| Sweat Shops Athletic | 334.4563333 | 10033.69 |
| Smith & Smith | 295.8818182 | 9764.10 |

One important thing to keep in mind is that if you include a full path in your file names, then you won't be able to move your output to another directory, as the path from the frame file to the contents and output files will be hard coded into your frame page. If you need to post your output to a web server, then this is important. It's better to leave the path out of the file names unless you can create them directly in their final location.

➤ REMOVING PROCEDURE LABELS

In the output above, the label "The MEANS Procedure" precedes each table. Chances are your end users don't care what procedure you used, and find this label annoying. To get rid of it, you can use the ODS NOPTITLE statement. This gets rid of any procedure labels in the body file, but leaves your user-defined titles in place. You'll also notice that the label "The Means Procedure" appears in the table of contents. To get rid of it there, you need to supply an alternate label with an ODS PROCLABEL statement. To implement both options, add the following lines of code:

```
ODS NOPTITLE;
ODS PROCLABEL "Hours Billed by Customer";
```

The NOPTITLE option will turn off all procedure titles in the body file until you reset the option by using an ODS PTITLE statement. However, the ODS PROCLABEL statement must be used before each procedure call. Even if you have two PROC PRINTs in a row, to modify both labels in the table of contents, you'll need two ODS PROCLABEL statements. Here is the resulting output after adding ODS NOPTITLE and ODS PROCLABEL.

| Table of Contents | | | | | | | | | | | | | | | | | | | |
|--|------|----------|------|-----|---------|-----|-------|-------------------------|-----|-------|-------------|-----|-------|----------------------|-----|-------|---------------|-----|-------|
| <i>Quarterly Report</i> | | | | | | | | | | | | | | | | | | | |
| <i>Hours Billed by Customer</i> | | | | | | | | | | | | | | | | | | | |
| Analysis Variable : Hours Worked | | | | | | | | | | | | | | | | | | | |
| <table border="1"> <thead> <tr> <th>Customer</th> <th>Mean</th> <th>Sum</th> </tr> </thead> <tbody> <tr> <td>IPO.com</td> <td>6.4</td> <td>598.5</td> </tr> <tr> <td>Bricks and Mortar, Inc.</td> <td>6.5</td> <td>575.0</td> </tr> <tr> <td>Virtual Co.</td> <td>5.7</td> <td>569.5</td> </tr> <tr> <td>Sweat Shops Athletic</td> <td>6.1</td> <td>658.0</td> </tr> <tr> <td>Smith & Smith</td> <td>5.7</td> <td>628.0</td> </tr> </tbody> </table> | | Customer | Mean | Sum | IPO.com | 6.4 | 598.5 | Bricks and Mortar, Inc. | 6.5 | 575.0 | Virtual Co. | 5.7 | 569.5 | Sweat Shops Athletic | 6.1 | 658.0 | Smith & Smith | 5.7 | 628.0 |
| Customer | Mean | Sum | | | | | | | | | | | | | | | | | |
| IPO.com | 6.4 | 598.5 | | | | | | | | | | | | | | | | | |
| Bricks and Mortar, Inc. | 6.5 | 575.0 | | | | | | | | | | | | | | | | | |
| Virtual Co. | 5.7 | 569.5 | | | | | | | | | | | | | | | | | |
| Sweat Shops Athletic | 6.1 | 658.0 | | | | | | | | | | | | | | | | | |
| Smith & Smith | 5.7 | 628.0 | | | | | | | | | | | | | | | | | |

➤ CREATING REPORTS TO IMPORT INTO WORD

Just as you can use ODS HTML to create output destined for the web, you can use ODS RTF to generate a file in Rich Text Format, which can be opened in Word (or other word processors). Again, the syntax is to add an ODS statement with a FILE option before your reporting procedure, and an ODS CLOSE statement after the end of your procedure. In this case, both ODS statements specify RTF as the output destination, and the filename has an ".rtf" extension. This example shows the output from a REPORT procedure, but is similar to the TABULATE example above.

```
ODS RTF FILE='myfilename.rtf';
* the REPORT code goes here ;
ODS RTF CLOSE;
```

| Education | Gender | Salary |
|----------------------|--------|----------|
| Not HS graduate | Female | \$4,449 |
| | Male | \$15,113 |
| HS graduate | Female | \$17,539 |
| | Male | \$33,419 |
| Some college | Female | \$22,730 |
| | Male | \$30,466 |
| Associates degree | Female | \$33,988 |
| | Male | \$40,690 |
| Bachelors degree | Female | \$43,862 |
| | Male | \$46,625 |
| Post-graduate degree | Female | \$45,000 |
| | Male | \$77,195 |

The output above shows what your SAS output looks like after it is opened in Word. The table of results is set up as a Word table, which means you can use all of the Word table formatting features to modify it if you like. To insert this table in another Word document, simple select the table and use copy/paste to drop it into another document.

➤ TITLES AND FOOTNOTES IN WORD

If you look closely at your ODS RTF output after it is opened in Word, you'll see that the titles and footnotes appear to be gray. If you print the document, or view it in Print Preview mode, the titles and footnotes will come out black. The reason for this is that ODS puts the titles into the Word document header, and the footnotes into the Word document footer. These parts of the document show up in gray when you are editing the document.

| <i>Example Title</i> | | |
|----------------------|---------------|---------------|
| Education | Gender | Salary |
| Not HS graduate | Female | \$4,449 |
| | Male | \$15,113 |
| HS graduate | Female | \$17,539 |
| | Male | \$33,419 |

If you don't like the way this looks, or if you need to use the Word header and footer for other titles and footnotes, then you can turn this behavior off. On the ODS RTF statement, you can add a BODYTITLE option to request that the titles and footnotes be made part of the body of the document. The syntax and output are as follows:

ODS RTF FILE='myfilename.rtf' BODYTITLE;

| <i>Example Title</i> | | |
|----------------------|---------------|---------------|
| Education | Gender | Salary |
| Not HS graduate | Female | \$4,449 |
| | Male | \$15,113 |
| HS graduate | Female | \$17,539 |

➤ ADJUSTING RTF PAGE MARGINS

There is one more issue that comes up with ODS RTF output that you may need to address. By default, the RTF output has margins of 0.25" when you open it in Word. This is generally a lot smaller than most users would like. To fix this, you can reset the margins after you open the document in Word. Another option is to customize your ODS style to fix the margins permanently. For instructions on how to do this, see the example on RTF margins in my paper "ODS Tips & Tricks", available at http://www.laurenhawksworth.com/pubs_current.htm.

For more information about creating RTF output, check out the following papers:

"Now There Is an Easy Way to Get to Word, Just Use PROC TEMPLATE, PROC REPORT, and ODS RTF," by Bob Hull, at www2.sas.com/proceedings/sugi26/p163-26.pdf.

"To ODS RTF and Beyond," by David Shannon, at www2.sas.com/proceedings/sugi27/p001-27.pdf.

➤ SENDING RESULTS TO PDF

If you need to deliver printable reports via e-mail or the web, you may want to try out the PDF destination. This file format can be viewed using a free utility called Acrobat Reader, which you can download from www.adobe.com.

The format can be viewed on many platforms, and its real strength is that your report will print out easily on many different printers, without any problems with margins and page breaks.

The code is quite simple. It's just like the RTF code, except you call ODS PDF, and the filename gets a “.pdf” extension.

```
ODS PDF FILE='myfilename.pdf';
* The REPORT code goes here ;
ODS PDF CLOSE;
```


| Education | Gender | Salary |
|----------------------|---------------|---------------|
| Not HS graduate | Female | \$4,449 |
| | Male | \$15,113 |
| HS graduate | Female | \$17,539 |
| | Male | \$33,419 |
| Some college | Female | \$22,730 |
| | Male | \$30,466 |
| Associates degree | Female | \$33,988 |
| | Male | \$40,690 |
| Bachelors degree | Female | \$43,862 |
| | Male | \$46,625 |
| Post-graduate degree | Female | \$45,000 |
| | Male | \$77,195 |

You will notice that the output produced by ODS PDF looks a lot like the output produced by ODS RTF. That's because the two ODS styles used by RTF and PDF are closely related. You can create any look you like in either PDF or RTF by switching to another style, or even creating your own custom style.

One warning, while PDF output is supported in SAS versions 8.0 and 8.1, the syntax is a little different, and the results are somewhat unreliable. Version 8.2 or higher is highly recommended for producing PDF output.

➤ CREATING PDF BOOKMARKS

By default, SAS creates bookmarks for your PDF output. Generally, the headings and subheadings in these bookmarks are similar to those that would be generated in an ODS HTML table of contents. To view these bookmarks, you may switch to this view in Adobe Acrobat. Use the Show Bookmarks command on the Windows pull-down menu. Below is an example of PDF bookmarks for some output from the RSREG procedure.



If you don't like these bookmarks, you have some options. First, you can use the ODS PROCLABEL statement to give them new labels, just like with the HTML table of contents. Second, you can get rid of them completely by adding the NOTOC option to your ODS PDF statement:

```
ODS PDF FILE='myfilename.pdf' NOTOC;
```

➤ ADJUSTING PDF PAGE MARGINS

Unlike the RTF destination, it is easy to change your margins for PDF output. You don't need to modify the ODS style definition. Instead, you can just use the system options LEFTMARGIN, RIGHTMARGIN, TOPMARGIN, and BOTTOMMARGIN. For example, to get 1 inch margins all around, use:

```
OPTIONS LEFTMARGIN=1in RIGHTMARGIN=1in TOPMARGIN=1in BOTTOMMARGIN=1in;
```

For more information on ODS PDF, check out the following paper in the SUGI 28 proceedings: "ODS PDF: It's not just for printing anymore!!" by Kevin Delaney.

➤ CREATING LANDSCAPE OUTPUT

With either RTF or PDF output, you may find that your output is too wide to fit on a standard portrait page. To rotate your output so you can use the wider landscape orientation, add the following system option before you call your output procedure.

```
OPTIONS ORIENTATION=LANDSCAPE;
```

➤ SENDING RESULTS TO EXCEL

There isn't an ODS EXCEL destination, or an ODS SPREADSHEET destination, but it's very easy to create an Excel file directly from SAS.

The way you do this is a cheat. Excel knows how to read HTML files, so you can open ODS HTML output directly from Excel. The tables, rows, and columns all transfer over correctly. You could always do a "Save As" to convert the file to Excel format, however there's a sneakier approach. In your ODS HTML statement, give the output file an ".xls" extension.

```
ODS HTML FILE='myfile.xls';
* the TABULATE code goes here ;
ODS HTML CLOSE;
```

| | A | B | C | D |
|----|-----------------------|----------|----------|----------|
| 1 | | | | |
| 2 | | | | |
| 3 | Average Salary | | | |
| 4 | | Male | Female | All |
| 5 | Education | | | |
| 6 | Not HS graduate | \$15,113 | \$4,449 | \$13,039 |
| 7 | HS graduate | \$33,419 | \$17,539 | \$28,464 |
| 8 | Some college | \$30,466 | \$22,730 | \$27,514 |
| 9 | Associates degree | \$40,690 | \$33,988 | \$38,057 |
| 10 | Bachelors degree | \$46,625 | \$43,862 | \$45,821 |
| 11 | Post-graduate degree | \$77,195 | \$45,000 | \$60,501 |

The above output shows what the file looks like when opened in Excel. Even though it's an HTML file, when you double-click on it to open the file, Excel automatically launches, and figures out how to handle the HTML. You can do a Save to ensure that the file is saved in true Excel format, but that's not necessary.

If you don't want the default colors and formatting, and prefer a simpler spreadsheet, try switching to the plain HTML destination by using "ODS PHTML" instead of "ODS HTML".

➤ SENDING RESULTS TO POWERPOINT

To get your SAS output into a PowerPoint slide, again ODS HTML is the best format to use. However, instead of opening the results directly from PowerPoint, you will be better off copying and pasting individual tables from your web browser into your PowerPoint slides.

This ensures that the tables are added in an appropriate size for your slide. If you open HTML directly from PowerPoint, it tends to convert the tables into extremely small tables on your slide, and resizing them is difficult.

The figure below shows a SAS table created by PROC TABULATE and output to HTML, as it looks when pasted onto a PowerPoint slide.

| Education | Male | Female | All |
|----------------------|----------|----------|----------|
| Not HS graduate | \$15,113 | \$4,449 | \$13,039 |
| HS graduate | \$33,419 | \$17,539 | \$28,464 |
| Some college | \$30,466 | \$22,730 | \$27,514 |
| Associates degree | \$40,690 | \$33,988 | \$38,057 |
| Bachelors degree | \$46,625 | \$43,862 | \$45,821 |
| Post-graduate degree | \$77,195 | \$45,000 | \$60,501 |


➤ CREATING OUTPUT DATASETS

While ODS is a great tool for generating attractive printed and electronic reports, there's also another facet of ODS worth consideration. The ODS OUTPUT destination allows you to capture virtually any output created by a SAS procedure. Before ODS, some procedures produced output datasets, and others did not. Even those procedures that did produce output datasets did not always include all of the procedure results in the output datasets. ODS OUTPUT was a big step forward.

To understand the ODS OUTPUT destination, you first need to understand output objects. Each time you create output from a SAS procedure, you are creating one or more output objects. The number and type of objects is determined by each SAS procedure. Generally, each table in your output is a separate output object. When you use the ODS OUTPUT destination, you specify which output object(s) you want. The syntax for ODS OUTPUT is very simple:

```
ODS OUTPUT <output object name>=<dataset name>;
* the procedure(s) go here;
ODS OUTPUT CLOSE;
```

The only thing that's tricky about this destination is figuring out the output object name for the results you want. A SAS procedure can generate a number of output objects. You need to figure out the name for the one you want. One of the simplest ways to do this is to run the procedure using the SAS Display Manager, and then view the results in the Results window. The following example shows the Results window after running the UNIVARIATE procedure.



The UNIVARIATE output for the variable Income is made up of five output objects: Moments, Basic Measures of Location and Variability, Tests for Location, Quantiles, and Extreme Observations. By looking at the output and looking at this window, it is possible to figure out which one contains the results you want for your dataset. Then, if you right-click on that object in the Results window and select Properties from the pop-up menu, you can look up the name of the output object.

For example, if you decide you want the results from the Basic Measures section of the output, the properties dialog will tell you that the name of that output object is BasicMeasures. To create a dataset called MyResults, the ODS syntax would be:

```
ODS OUTPUT BasicMeasures=MyResults;
* the UNIVARIATE procedure goes here;
ODS OUTPUT CLOSE;
```

The resulting dataset looks like this (PROC PRINT output):

| Obs | Var Name | Loc Measure | LocValue | VarMeasure | VarValue |
|-----|----------|-------------|----------|---------------------|-----------|
| 1 | Income | Mean | 32487.93 | Std Deviation | 26143 |
| 2 | Income | Median | 25700.00 | Variance | 683470675 |
| 3 | Income | Mode | 99999.00 | Range | 98999 |
| 4 | Income | | - | Interquartile Range | 30041 |

To see how this relates to the procedure output, here are the original UNIVARIATE results for Basic Measures. In this case, the output dataset and the original procedure output are very similar in structure. That is not always the case. Some procedures produce output datasets very different in structure from the printed results. You'll need to spend some time getting familiar with each procedure's output datasets.

| Basic Statistical Measures | | | |
|----------------------------|----------|---------------------|-----------|
| Location | | Variability | |
| Mean | 32487.93 | Std Deviation | 26143 |
| Median | 25700.00 | Variance | 683470675 |
| Mode | 99999.00 | Range | 98999 |
| | | Interquartile Range | 30041 |

The above example created a single output dataset. However, you can create more than one at a time. Just list a series of <output object name>=<dataset name> pairs in your ODS OUTPUT statement. Also, it's possible to combine the output objects from more than one procedure call into a single output dataset. For information on how to do this, read about the MATCH_ALL and PERSIST options in SAS OnlineDoc or SAS Help.

One word of warning about output datasets: if you don't want printed output, you can't use the NOPRINT option. This option stops the output from being created, so there's no output object for ODS OUTPUT to capture. To turn off output to the standard listing destination (the screen or a file), you need to use an ODS LISTING CLOSE statement. Once you're done creating the output dataset, you can turn listing output back on using an ODS LISTING statement.

➤ IN-LINE FORMATTING

One of the best new ODS features is also the least-documented of the new features. In-line formatting is a powerful tool to let you insert custom formatting anywhere there is text in your ODS output. This technique does not work for listing output; in fact it will make a mess of your listing output. But for HTML, RTF, and PDF, it's very powerful.

As a simple example, we'll use in-line formatting to create superscript footnote references in a table. We'll indicate a couple of references in the body of the table by adding a superscript 1 and 2 (actually, ¹ and ²). Then, we'll use the same superscripts in two footnotes below the table.

The first step is to turn on in-line formatting. This is done by specifying a special character to be used to identify in-line formatting sequences. This needs to be a character that won't be needed anywhere else in the output. One popular character to use is the caret (^), but the choice is up to you. The syntax for identifying the character is:

```
ODS ESCAPECHAR='^';
```

The next step is to find a way to apply the formatting to the appropriate text. In this example, we'll add the footnotes to selected row headings by adding the formatting characters to the PROC FORMAT labels for each row. To indicate a superscript, the escape character is followed by a special command in brackets. For a superscript, the command is the keyword "super", followed by a space, followed by the text to superscript.

```
proc format;
  value txft 1='Control'
    2='Low Dose^{super 1}'
    3='High Dose^{super 2}';
run;
```

This format, when applied to the row variable in a table, creates the following output:

| Treatment Group | Variable | Label | Mean | Std Dev |
|------------------------|----------|-----------------------------|-------|---------|
| Control | postsbp | Post-Treatment Systolic BP | 112.4 | 7.6 |
| | postdbp | Post-Treatment Diastolic BP | 84.3 | 11.6 |
| Low Dose ¹ | postsbp | Post-Treatment Systolic BP | 114.1 | 8.4 |
| | postdbp | Post-Treatment Diastolic BP | 83.1 | 10.5 |
| High Dose ² | postsbp | Post-Treatment Systolic BP | 109.4 | 7.7 |
| | postdbp | Post-Treatment Diastolic BP | 69.0 | 13.9 |

Now, to add the appropriate footnotes, the same in-line formatting command is applied to the text in the FOOTNOTE statements:

```
footnote "^{super 1}2 mL/kg.";
footnote2 "^{super 2}5 mL/kg";
```

| Treatment Group | Variable | Label | Mean | Std Dev |
|------------------------|----------|-----------------------------|-------|---------|
| Control | postsbp | Post-Treatment Systolic BP | 112.4 | 7.6 |
| | postdbp | Post-Treatment Diastolic BP | 84.3 | 11.6 |
| Low Dose ¹ | postsbp | Post-Treatment Systolic BP | 114.1 | 8.4 |
| | postdbp | Post-Treatment Diastolic BP | 83.1 | 10.5 |
| High Dose ² | postsbp | Post-Treatment Systolic BP | 109.4 | 7.7 |
| | postdbp | Post-Treatment Diastolic BP | 69.0 | 13.9 |

¹2 mL/kg.
²5 mL/kg

The resulting output (above) now has footnotes that match the references in the table. However, the footnotes look a little odd because they are so much bigger than the table text. To fix this, we can adjust the font size used in the footnotes via another in-line formatting command. The style command uses a capital “S” followed by an equal sign and then brackets containing a style attribute.

```
footnote "^{font_size=2}{super 1}2 mL/kg.";
footnote2 "^{font_size=2}{super 2}5 mL/kg";
```

The new output looks like this:

| Treatment Group | Variable | Label | Mean | Std Dev |
|------------------------|----------|-----------------------------|-------|---------|
| Control | postsbp | Post-Treatment Systolic BP | 112.4 | 7.6 |
| | postdbp | Post-Treatment Diastolic BP | 84.3 | 11.6 |
| Low Dose ¹ | postsbp | Post-Treatment Systolic BP | 114.1 | 8.4 |
| | postdbp | Post-Treatment Diastolic BP | 83.1 | 10.5 |
| High Dose ² | postsbp | Post-Treatment Systolic BP | 109.4 | 7.7 |
| | postdbp | Post-Treatment Diastolic BP | 69.0 | 13.9 |

¹2 mL/kg.
²5 mL/kg

Font sizes and superscripting are just a two of the many text attributes you can control via in-line formatting and style attributes.

For more information on in-line formatting, see the following SUGI papers:

“Changes & Enhancements for ODS by Example (through Version 8.2),” by Sandy McNeill, at
<http://www2.sas.com/proceedings/sugi26/p002-26.pdf>

“What’s New in the Output Delivery System, Version 9.0,” by Sandy McNeill, at
<http://www2.sas.com/proceedings/sugi27/p128-27.pdf>

➤ CONCLUSIONS

This concludes our quick tour of the Output Delivery System. Hopefully you now have a basic idea of what options are available to you. Since this was only an introduction, as you start to use these techniques, be sure to explore the SAS documentation, as well as the SUGI papers referenced previously, for more in-depth discussion.

➤ ACKNOWLEDGEMENTS

SAS is a registered trademark or trademark of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

➤ CONTACTING THE AUTHOR

Please direct any questions or feedback to the author at: info@laurenhaworth.com